

Lab 10: Random effects models in SAS
STT 422: Summer, 2004
Vince Melfi

There are various ways to fit random effects ANOVA models in SAS, including **proc glm**, **proc varcomp**, and **proc mixed**. For the models that we'll be interested in, **proc glm** will suffice, but we'll also use **proc mixed**, since it is a better choice for more complicated random and mixed effects models.

The data

A company has a large number of personnel officers, who rate job applicants on a scale of 0 to 100. In order to assess the variability in the ratings given by different officers, a study was done. Five officers were chosen at random from all the officers. Twenty job applicants were randomly assigned to the five officers, with four per officer. The data are the ratings given to the twenty applicants. The following SAS program reads in the data, prints it, computes the mean ratings separately for each officer, and plots the data by officer.

```
data personnel;
  infile 'c:\classes\summer04\422\data\ratings.dat';
  input rating officer;

proc print data = personnel;

proc means data = personnel;
  var rating;
  by officer;

symbol1 v = circle;
proc gplot data = personnel;
  plot rating * officer;

run;
```

Fitting a random effects model using proc glm

The **random** statement lets SAS know that we want to fit a random effects model using **proc glm**. Here's how to do this for the ratings data.

```
proc glm data = personnel;
  class officer;
  model rating = officer;
  random officer;

run;
```

The output is similar to the output from **proc glm** when fitting fixed effects models. The only difference is the inclusion of the expected mean squares after the analysis of variance table. This is needed in order to compute the estimate of σ_μ^2 . In this example, the expected mean square for the model is $\sigma^2 + 4\sigma_\mu^2$, so the estimate of σ_μ^2 would be the mean square for the model minus the mean square for error divided by 4:

$$\hat{\sigma}_\mu^2 = \frac{394.925000 - 73.283333}{4} \approx 80.409.$$

Remember that the F statistic given in the analysis of variance table is testing the null hypothesis $H_0: \sigma_\mu^2 = 0$. In this case the p-value is 0.0068.

Fitting a random effects model using **proc mixed**

The **proc mixed** procedure fits a wide variety of random and mixed effects models, allowing for a potentially complicated covariance structure (lack of independence) in the data. The underlying details of how the model is fit differ from **proc glm**, but the results are the same in our relatively simple example. We'll use it to fit the one-way random effects model to the ratings data.

```
proc mixed data = personnel;  
  class officer;  
  model rating =;  
  random officer;  
  
run;
```

The syntax is similar to that of **proc glm**, with one important difference. In the **model** statement, only *fixed* effects are listed on the right hand side. Random effects are not listed, but are given in the **random** statement.

The output is quite different. For our purposes, the most interesting portion is the **covariance parameter estimates** section, which gives estimates of σ^2 and σ_μ^2 directly.